
Abstract

In a case study we investigate whether off the shelf higher-order theorem provers and model
generators can be employed to automate reasoning in and about quantified multimodal logics.
In our experiments we exploit the new TPTP infrastructure for classical higher-order logic.

1 Introduction

This paper presents a case study in quantified multimodal logics. An interesting aspect of this
case study is that off the shelf theorem provers and model generators for simple type theory, that
is, classical higher-order logic, are employed to automateproblems in quantified multimodal log-
ics, that is, non-classical logics. This is enabled by our recent embedding of normal quantified
multimodal logics in simple type theory [8, 10], which is sound and complete [10]. Interestingly,
not only reasoningwithin various non-classical logics can be automated this way but also rea-
soningabout them. For example, the equivalence between different properties of accessibility
relations and their associated multimodal axioms can be proved automatically.

Generally, it seems that proof assistants and automated reasoning systems for classical higher-
order logics can serve — ideally in combination — as a fruitful basis for the exploration and the
modeling of normal multimodal logics, for the investigation of their meta-properties, and proba-
bly even for the application of these logics to real world problems. In fact, normal multimodal log-
ics are simply fragments of classical higher-order logics with specific computational properties.
Detecting and characterizing these fragments, in particular decidable ones, can be very beneficial
for the higher-order reasoners themselves, since this may stimulate significant improvements of
the reasoning procedures used within these reasoners.

The modeling of a specific propositional or quantified multimodal logic of interest in our
framework is straightforward and not very time consuming. This is in contrast with the usual
situation in the area of quantified and propositional multimodal logics, where a vast amount of
specific calculi have been developed and published, while very few of them have actually been
implemented and applied. The main reason is that the effective implementation of these calculi is
usually non-trivial.

The paper is structured as follows: In Section 2 we present some preliminaries and our em-
bedding of quantified multimodal logic into simple type theory. The new content, our small but
interesting case study, is presented in Section 3 and an outlook is presented in Section 4.1

1Initially our plan was to extend the case presented here to propositional and quantified temporal logics. A study
in (quantified) temporal logics will thus remain future work.)



3

2 QML as a fragment ofST T

ST T [15] is based on the simply typedλ-calculus. The setT of simple types is usually freely
generated from a set of basic types{o, ι} (whereo is the type of Booleans andι is the type of
individuals) using the function type constructor�. Instead of{o, ι} we here consider a set of
base types{o, ι, µ}, providing an additional base typeµ (the type of possible worlds).

The simple type theory languageST T is defined by (α, β ∈ T ):

s, t ::= pα | Xα | (λXα sβ)α�β | (sα�β tα)β | (¬o�o so)o |

(so ∨o�o�o to)o | (sα =α�α�o tα)o | (Π(α�o)�o sα�o)o

pα denotes typed constants andXα typed variables (distinct frompα). Complex typed terms
are constructed via abstraction and application. Our logical connectives of choice are¬o�o,
∨o�o�o, =α�α�o andΠ(α�o)�o (for each typeα).2 From these connectives, other logical con-
nectives can be defined in the usual way (e.g.,∧ and =⇒ ). We often use binder notation∀Xα s

for Π(α�o)�o(λXα so). We assume familiarity withα-conversion,β- andη-reduction, and the
existence ofβ- andβη-normal forms. Moreover, we obey the usual definitions of free variable
occurrences and substitutions.

The semantics ofST T is well understood and thoroughly documented in the literature [2, 3,
7, 19]. The semantics of choice for our work is Henkin semantics.

QML has been studied by Fitting [16] (further related work is available by Blackburn and
Marx [13] and Braüner [14]). In contrast to Fitting we are here not interested only inS5structures
but in the more general case ofK from which more constrained structures (such asS5) can be
easily obtained. First order quantification can be constantdomain or varying domain. Below we
only consider the constant domain case: every possible world has the same domain. Like Fitting,
we keep our definitions simple by not having function or constant symbols. While Fitting [16]
studies quantified monomodal logic, we are interested in quantified multimodal logic. Hence,
we introduce multiple2r operators for symbolsr from an index setS. The grammar for our
quantified multimodal logicQML hence is

s, t ::= P | k(X1, . . . , Xn) | ¬ s | s ∨ t | ∀X s | ∀P s | 2r s

whereP denotes propositional variables,X,X i denote first-order (individual) variables, andk
denotes predicate symbols of any arity. Further connectives, quantifiers, and modal operators
can be defined as usual. We also obey the usual definitions of free variable occurrences and
substitutions.

Fitting introduces three different notions of Kripke semantics forQML: QS5π−, QS5π, and
QS5π+. In our work [10] we study related notionsQKπ−, QKπ, andQKπ+ for a modal context
K , and we support multiple modalities.

ST T is an expressive logic and it is thus not surprising thatQML can be elegantly mod-
eled and even automated as a fragment ofST T . The idea of the encoding, calledQMLSTT ,
is simple. Choose typeι to denote the (non-empty) set of individuals and we reserve asec-
ond base typeµ to denote the (non-empty) set of possible worlds. The typeo denotes the set

2This choice is not minimal (from=α�α�o all other logical constants can already be defined [4]). It useful though
in the context of resolution based theorem proving.

of truth values. Certain formulas of typeµ � o then correspond to multimodal logic expres-
sions. The multimodal connectives¬ , ∨ , and2 , becomeλ-terms of types(µ � o) � (µ � o),
(µ � o) � (µ � o) � (µ � o), and(µ � µ � o) � (µ � o) � (µ � o) respectively.

Quantification is handled as inST T by modeling∀X p asΠ(λX .p) for a suitably chosen
connectiveΠ. Here we are interested in defining two particular modalΠ-connectives:Πι, for
quantification over individual variables, andΠµ�o, for quantification over modal propositional
variables that depend on worlds. They become terms of type(ι � (µ � o)) � (µ � o) and
((µ � o) � (µ � o)) � (µ � o) respectively.

TheQMLSTT modal operators¬ , ∨ ,2,Πι, andΠµ�o are now simply defined as follows:

¬ (µ�o)�(µ�o) = λφµ�o λWµ ¬(φW )

∨ (µ�o)�(µ�o)�(µ�o) = λφµ�o λψµ�o λWµ φW ∨ ψW

2 (µ�µ�o)�(µ�o)�(µ�o) = λRµ�µ�o λφµ�o λWµ ∀Vµ ¬(RW V ) ∨ φ V

Πι
(ι�(µ�o))�(µ�o) = λφι�(µ�o) λWµ ∀Xι φXW

Π
µ�o

((µ�o)�(µ�o))�(µ�o) = λφ(µ�o)�(µ�o) λWµ ∀Pµ�o φP W

Further operators can be introduced as usual, for example,⊤ = λWµ ⊤,⊥ = ¬ ⊤, ∧ =
λφ, ψ ¬ (¬φ ∨ ¬ψ), ⊃= λφ, ψ ¬φ ∨ ψ, ⇔ = λφ, ψ (φ ⊃ ψ) ∧ (ψ ⊃ φ), 3 =
λR, φ ¬ (2R (¬φ)), Σι = λφ ¬Πι(λX ¬φX), Σµ�o = λφ ¬Πµ�o(λP ¬φP ).

For definingQMLSTT -propositions we fix a setIVSTT of individual variables of typeι, a set
PVSTT of propositional variables3 of typeµ � o, and a setSYMSTT of n-ary (curried) predicate
constants of typesι � . . . � ι

︸ ︷︷ ︸

n

� (µ � o). Moreover, we fix a setSSTT of accessibility relation

constants of typeµ � µ � o. QMLSTT -propositions are now defined as the smallest set of
ST T -terms for which the following hold:

• if P ∈ PVSTT , thenP ∈ QMLSTT

• if Xj ∈ IVSTT (j = 1, . . . ,n) andk ∈ SYMSTT , then(k X1 . . . Xn) ∈ QMLSTT

• if φ, ψ ∈ QMLSTT , then¬ φ ∈ QMLSTT andφ ∨ ψ ∈ QMLSTT

• if r ∈ SSTT andφ ∈ QMLSTT , then2 r φ ∈ QMLSTT .

• if X ∈ IVSTT andφ ∈ QMLSTT , thenΠι(λX φ) ∈ QMLSTT

• if P ∈ PVSTT andφ ∈ QMLSTT , thenΠµ�o(λP φ) ∈ QMLSTT

We write2r φ for 2 r φ, ∀Xι φ for Πι(λXι φ), and∀Pµ�o φ for Πµ�o(λPµ�o φ).

Note that the defining equations for ourQML modal operators are themselves formulas in
simple type theory. Hence, we can expressQML formulas in a HO-ATP elegantly in the usual
syntax. For example,2r ∃Pµ�o P is aQMLSTT proposition; it has typeµ � o.

Validity of QMLSTT propositions is defined in the obvious way: aQML-propositionφµ�o

is valid if and only if for all possible worldswµ we havew ∈ φµ�o, that is, if and only ifφµ�owµ

holds. Hence, the notion of validity is modeled via the following equation:

mvalid = λφµ�o ∀Wµ φW

3Note that the denotation of propositional variables depends on worlds.

Now we can formulate proof problems inQMLSTT , e.g., mvalid2r ∃Pµ�o P . Using rewriting
or definition expanding, we can reduce such proof problems tocorresponding statements con-
taining only the basic connectives¬ , ∨ , =, Πι, andΠµ�o of ST T . In contrast to the many
other approaches no external transformation mechanism is required. For our example we get
∀Wµ ∀Yµ ¬(rW Y ) ∨ (¬∀Xµ�o ¬(X Y )). It is easy to check that this term is valid in Henkin
semantics: putX = λYµ ⊤.

We have proved soundness and completeness for this embedding [10].

Theorem 2.1 (Soundness) If |=ST T (validsµ�o) then|=QKπ s.

Theorem 2.2 (Completeness) If |=QKπ s then|=ST T (validsµ�o).

These results illustrate the natural correspondence between QKπ models and Henkin models.
Moreover, we get the following corollaries.

Corollary 2.3 The reduction of our embedding to propositional quantified multimodal logics
(which only allow quantification over propositional variables) is sound and complete.

Corollary 2.4 The reduction of our embedding to first-order multimodal logics (which only allow
quantification over individual variables) is sound and complete.

Corollary 2.5 The reduction of our embedding to propositional multimodallogics (no quantifi-
cation) is sound and complete.

In the remainder of the paper we will usually omit type information. It is sufficient to remember
that states are of typeµ, multimodal propositions of typeµ � o, and accessibility relations of
typeµ � µ � o.

3 A Case Study

In this section we apply off the shelf reasoning systems for simple type theory to simple problems
in and about quantified multimodal logics. Many of these examples have been adapted from
Goldblatt’s textbook [18]. Our problems are encoded in the new TPTP THF syntax [11] and our
experiments exploit the new higher-order TPTP infrastructure [24]. The reasoning systems we
apply are LEO-II (version 0.99a) [12], TPS (version 3.0) [1]and IsabelleP (version 2008) and
IsabelleM (version 2008) of the Isabelle proof assistant [22].

LEO-II is a resolution based higher-order ATP system. LEO-II is implemented in Objective
Caml, and is freely available under a BSD-like license. LEO-II is designed to cooperate with spe-
cialist systems for fragments of higher-order logic. Currently, LEO-II is capable of cooperating
with the first-order ATP systems E, SPASS, and Vampire. LEO-II directly parses THF0 input and
communicates with the cooperating first-order ATP system using TPTP standards.

TPS is a higher-order theorem proving system that has been developed under the supervision
of Peter B. Andrews since the 1980s. Theorems can be proved either interactively or automat-
ically. In TPS there are flags that can be set to affect the behavior of automated search. The
automated TPS used for solving THF problems uses two different collections of flags, in modes
called MS98-FO-MODE and BASIC-MS04-2-MODE. As the two modes have quite different
capabilities, they are run in competition parallel as a simple way of obtaining greater coverage.

Isabelle is normally used interactively. A fully automaticversion, called IsabelleP, has been
implemented using strategy scheduling of the nine automatic tactics simp, blast, auto, metis, fast,
fastsimp, best, force, and meson. While it was probably never intended to use Isabelle as a fully
automatic system, this simple automation provides useful capability. The ability of Isabelle to
find (counter-) models using the refute tactic has also been integrated into an automatic system,
called IsabelleM.

The first interesting question clearly is whether our embedding QMLSTT of quantified mul-
timodal logic in simple type theory is consistent. The file inAttachment A contains the encoding
of our embedding in TPTP THF syntax. The file also contains some further notions and concepts
as will be introduced later in this paper. When applying the model generator IsabelleM to this in-
put file, we receive in 0.7 seconds the answer that this set of definitions and axioms is satisfiable.
Thus, IsabelleM confirms that our theory is consistent.

All experiments were conducted with the SystemOnTPTP interface [23] which provides on-
line access to the theorem provers running locally at computers at University of Miami. These
provers were called remotely with a timeout of 200s.

Problem IsabelleP LEO-II TPS IsabelleM
(1) THM(40.0) THM(0.0) THM(0.3) TMO
(2) THM(105.6) THM(0.0) THM(0.4) TMO
(3) THM(149.6) THM(0.0) THM(0.2) TMO
(4) THM(123.8) THM(0.0) THM(0.3) TMO
(5) TMO THM(0.1) THM(0.3) TMO
(6) TMO THM(0.1) THM(0.2) TMO

Table 1:

3.1 Reasoning within Multimodal Logics

We study whether some simple textbook examples can be automatically proved.

Example 3.1 (cf. [18], Exercise 1.4) The following statements are valid (for all accessibility
relationsr).

2r ⊤ (1)

∀A,B 2r (A ⊃ B) ⊃ (2r A ⊃ 2r B) (2)

∀A,B 3r (A ⊃ B) ⊃ (2r A ⊃ 3r B) (3)

∀A,B 2r (A ⊃ B) ⊃ (3r A ⊃ 3r B) (4)

∀A,B 2r (A ∧ B) ⇔ (2r A ∧ 2r B) (5)

∀A,B 3r (A ∨ B) ⇔ (3r A ∨ 3r B) (6)

The THF encoding of Example (3) is given in Appendix B.

The results of our experiment are given in Table 1. The table entries are to be interpreted as
follows: THM(40.0) says that a problem was classified as ’theorem’ within 40.0 seconds. TMO
says that the prover was killed due to reaching the ’timeout’limit of 200 seconds. UKN specifies
that the prover stopped proof search but without generatingany classification result. CSA(0.8)
(cf. Table 2) says that a problem was classified as counter-satisfiable in 0.8 seconds.

The next set of examples is more challenging. These examplesare about counter-satisfiable
statements (in basic multimodal logic K). Proving them is non-trivial since the synthesis of ap-
propriate accessibility relationsR and propositionsA,B are required.4

4These examples are actually already belonging to the next subsection, since they rather illustrate the poten-
tial of our systems for reasoning about multimodal logics than reasoning within them — the synthesis of concrete
accessibility relations, for instance, is to the best of ourknowledge not supported in traditional, direct approaches.

Example 3.2 (cf. [18], Exercise 1.5)The following propositions are not valid.

∀A 2r A ⊃ A (7)

∀A 2r A ⊃ 2r 2r A (8)

∀A,B 2r (A ⊃ B) ⊃ (2r A ⊃ 3r B) (9)

3r ⊤ (10)

∀A 3r A ⊃ 2r A (11)

∀A,B 2r (2r A ⊃ B) ∨ 2r (2r B ⊃ A) (12)

∀A,B 2r (A ∨ B) ⊃ (2r A ∨ 2r B) (13)

∀A 2r (2r A ⊃ A) ⊃ 2r A (14)

The encoding of these examples is analogous to the encoding of the previous ones. However,
for each problem we also formulate a second question which asks whether there is an accessibility
relationr such that the proposition is not valid. For instance, for Example 8 this second question,
called problem (8a) below, is encoded as follows:

%---- include the definitions for quantified multimodal logic
include(’QML.ax’).
%---- conjecture statement
thf(conj,conjecture,(

? [R: mu > mu > $o] :
~ ( mvalid
@ ( mall_prop

@ ^ [A: mu > $o] :
( mimpl @ ( mbox @ R @ A )

@ ( mbox @ R @ ( mbox @ R @ A ) ) ) ) ) )).

The results of this experiment are given in Table 2. Note thatour provers generate the ex-
pected results for examples (7),(7a),(9),(9a),(10),(10a),(14), and (14a), that is, the model finder
IsabelleM finds a countermodel for Examples (7),(9),(10), and (14) and even the provers can
show that there exists countermodels by proving the corresponding statements (7a),(9a),(10a),
and (14a).

For the remaining problems the task is more challenging though and we will illustrate this
with the help of Example (8), the transitivity axiom. This axiom is not valid and correctly none
of the theorem provers signals that a proof can be found. Unfortunately, the model generator
IsabelleM does not generate a countermodel either. We thus try to prove (8a) which states that
there is an accessibility relationR such that the transitivity axiom is invalidated.

∃R (¬valid ∀A (2RA ⊃ 2R 2RA))

We would expect that this statement can be quickly proved when instantiating variableR with a
non-transitive relation. For good reasons our provers failto do so. In fact, our provers also fail to
prove the related statement

∃R ¬(transitive R)

Problem IsabelleP LEO-II TPS IsabelleM
(7) TMO UKN TMO CSA(1.6)
(7a) THM(79.6) THM(62.4) THM(0.2) TMO
(8) UKN UKN TMO TMO
(8a) UKN TMO TMO CSA(1.6)
(9) UKN TMO TMO CSA(1.7)
(9a) UKN THM(64.9) THM(0.2) TMO
(10) UKN UKN TMO CSA(1.5)
(10a) THM(78.9) THM(63.4) THM(0.2) TMO
(11) UKN UKN TMO TMO
(11a) TMO TMO UKN CSA(1.6)
(12) TMO UKN TMO TMO
(12a) UKN TMO TMO CSA(1.8)
(13) UKN UKN TMO TMO
(13a) UKN TMO TMO CSA(1.7)
(14) UKN TMO TMO CSA(1.6)
(14a) UKN THM(72.5) THM(8.6) TMO

Table 2:

The reason is that without further assumptions this statement is not a theorem. We have neither
assumed the axiom of infinity nor that there exist at least twodifferent possible worlds. Hence,
our domain of possible worlds may well just consist of a single worldw in which case a non-
transitive accessibility relation cannot be provided. It is thus not surprising that IsabelleM finds a
countermodel to statement (8a). Unfortunately, it does notsucceed though to find a countermodel
to the original problem.

3.2 Reasoning about Multimodal Logics

The literature on modal logics is full of theorems and student exercises illustrating the relation-
ships between properties of accessibility relations and corresponding axioms (respectively axiom
schemata). Therefore, the question that interests us next is whether such meta-theoretic results
can be automatically proved by our higher-order reasoners.We again study the examples given
in Goldblatt’s textbook [18]. First we encode various accessibility relation properties in simple
type theory.

reflexive = λR ∀S RS S (15)

symmetric = λR ∀S, T (RS T ) ⇒ (RT S) (16)

serial = λR ∀S ∃T (RS T ) (17)

transitive = λR ∀S, T, U ((RS T ) ∧ (RT U)) ⇒ (RS U) (18)

Euclidean = λR ∀S, T, U ((RS T ) ∧ (RS U)) ⇒ (RT U) (19)

partially_functional = λR ∀S, T, U ((RS T ) ∧ (RS U)) ⇒ (T = U) (20)

functional = λR ∀S ∃T (RS T ) ∧ ∀U (RS U) ⇒ (T = U) (21)

weakly_dense = λR ∀S, T (RS T ) ⇒ ∃U (RS U) ∧ (RU T ) (22)

weakly_connected= λR ∀S, T, U ((RS T ) ∧ (RS U)) ⇒

((RT U) ∨ (T = U) ∨ (RU T )) (23)

weakly_directed = λR ∀S, T, U ((RS T ) ∧ (RS U)) ⇒

∃V (RT V ) ∧ (RU V ) (24)

The formulation of these accessibility relation properties in THF syntax is straightforward,
for example,5

thf(mtransitive,definition,
( mtransitive
= ( ^ [R: mu > mu > $o]: ! [S: mu,T: mu,U: mu]:

(((R @ S @ T) & (R @ T @ U)) => (R @ S @ U))))).

The corresponding axioms are given next.

∀A 2r A ⊃ A (25)

∀A A ⊃ 2r 3r A (26)

∀A 2r A ⊃ 3r A (27)

∀A 2r A ⊃ 2r 2r A (28)

∀A 3r A ⊃ 2r 3r A (29)

∀A 3r A ⊃ 2r A (30)

∀A 3r A ⇔ 2r A (31)

∀A 2r 2r A ⊃ 2r A (32)

∀A,B 2r (A ∧ (2r A ⊃ B)) ∨ 2r (B ∧ (2r B ⊃ A)) (33)

∀A 3r 2r A ⊃ 2r 3r A (34)

5We have decided to use a prefix ’m’ for all these example to avoid clashes of these definitions with inbuilt
concepts in our reasoners using the same name.

Problem IsabelleP LEO-II TPS IsabelleM
(15)⇒ (25) THM(90.0) THM(0.0) THM(0.3) TMO
(16)⇒ (26) THM(113.7) THM(0.0) THM(0.2) TMO
(17)⇒ (27) THM(102.1) THM(0.0) THM(0.3) TMO
(18)⇒ (28) THM(123.2) THM(0.0) THM(8.1) TMO
(19)⇒ (29) THM(131.7) THM(0.0) THM(0.6) TMO
(20)⇒ (30) THM(125.8) THM(0.0) THM(1.1) TMO
(21)⇒ (31) THM(156.3) THM(0.0) TMO TMO
(22)⇒ (32) THM(105.1) THM(0.0) THM(1.1) TMO
(23)⇒ (33) THM(42.6) THM(0.0) THM(0.2) TMO
(24)⇒ (34) THM(178.1) THM(0.0) THM(0.2) TMO
(15)⇐ (25) THM(75.3) THM(0.0) THM(0.2) TMO
(16)⇐ (26) UKN TMO THM(0.5) TMO
(17)⇐ (27) THM(89.6) THM(0.0) THM(0.2) TMO
(18)⇐ (28) TMO TMO TMO TMO
(19)⇐ (29) UKN TMO TMO TMO
(20)⇐ (30) TMO THM(6.2) THM(47.9) TMO
(21)⇐ (31) TMO TMO THM(151.3) TMO
(22)⇐ (32) TMO TMO TMO TMO
(23)⇐ (33) UKN TMO TMO TMO
(24)⇐ (34) THM(85.1) THM(0.1) THM(148.5) TMO

Table 3:

Example 3.3 (Correspondence of accessibility relation properties and axioms)
Propertyk (k = (15), . . . , (24)) holds for accessibility relationr if and only if the corresponding
axiom (k+10) is valid.

As an example we present the encoding of problem (18)⇐ (28) in Appendix C. The outcome
of this experiment is summarized in Table 3. Except for 4 out of the 20 problems our provers
collectively are able to come up with the expected results.

3.3 Epistemic Reasoning

In this section we encode epistemic reasoning problems in our framework. The examples are
taken from Baldoni [5].

Example 3.4 (Epistemic reasoning: The friends puzzle) (i) Peter is a friend of John, so if Pe-
ter knows that John knows something then John knows that Peter knows the same thing. (ii) Peter
is married, so if Peter’s wife knows something, then Peter knows the same thing. John and Peter
have an appointment, let us consider the following situation: (a) Peter knows the time of their
appointment. (b) Peter also knows that John knows the place of their appointment. Moreover, (c)
Peter’s wife knows that if Peter knows the time of their appointment, then John knows that too
(since John and Peter are friends). Finally, (d) Peter knowsthat if John knows the place and the
time of their appointment, then John knows that he has an appointment.

From this situation we want to prove (e) that each of the two friends knows that the other one
knows that he has an appointment.

For the three persons in this puzzle, John, Peter and Peter’swife, we introduce corresponding
accessibility relationsjohn, peter, andwife(peter). For this example it is sufficient to require the
S4 axioms, that is, a knowledge or truth axioms (cf. 25) and positive introspection axioms (cf. 28)
for each accessibility relation. Alternatively, we here require that the accessibility relationsjohn,
peter, andwife(peter)are reflexive and transitive.

%----
thf(peter,type,( peter: mu > mu > $o )).
thf(john,type,( john: mu > mu > $o )).
thf(wife,type,(wife: ( mu > mu > $o ) > mu > mu > $o )).
%----
thf(refl_peter,axiom,( mreflexive @ peter )).
thf(refl_john,axiom,( mreflexive @ john )).
thf(refl_wife_peter,axiom,( mreflexive @ ( wife @ peter ))).
thf(trans_peter,axiom,( mtransitive @ peter )).
thf(trans_john,axiom,( mtransitive @ john)).
thf(trans_wife_peter,axiom,( mtransitive @ ( wife @ peter ))).

It is clear that the following S4 axioms are implied by these conditions on the accessibility rela-
tions (i ranges overwife(peter), peter, andjohn).

K(i) = ∀A 2i A ⊃ A T (i) = ∀A 2i A ⊃ 2i 2i A

We again use our reasoners to verify this. We only check this for K(wife(peter)) and T(wife(peter));
the results are given in Table 4.

Next, we encode the facts from the puzzle. For (i) we provide apersistence axiom and for (ii)
an inclusion axiom:

∀A 2peter2johnA ⊃ 2john 2peterA (35)

∀A 2wife(peter)A ⊃ 2peterA (36)

Problem IsabelleP LEO-II TPS IsabelleM
K(wife(peter)) THM(89.4) THM(0.0) THM(0.3) TMO
T(wife(peter)) THM(178.3) THM(0.0) TMO TMO
Example 3.4 TMO THM(0.1) TMO TMO
Example 3.5 UKN THM(0.3) TMO TMO

Table 4:

Finally, the facts (a)-(d) and the conclusion (e) are encoded as follows (time, place, and appoint-
ment are propositional constants, that is, constants of type ι � o in our framework):

2petertime (37)

2peter2john place (38)

2wife(peter)(2petertime ⊃ 2john time) (39)

2peter2john (place∧ time ⊃ appointment) (40)

2john 2peterappointment∧ 2peter2john appointment (41)

The THF encoding of the entire example is presented in Appendix D. Table 4 shows the results
of our experiment.

In the modeling of our next example we also follow Baldoni.

Example 3.5 (Wise men puzzle) Once upon a time, a king wanted to find the wisest out of his
three wisest men. He arranged them in a circle and told them that he would put a white or a
black spot on their foreheads and that one of the three spots would certainly be white. The three
wise men could see and hear each other but, of course, they could not see their faces reflected
anywhere. The king, then, asked to each of them to find out the color of his own spot. After a
while, the wisest correctly answered that his spot was white.

The encoding of the example is presented in Appendix E and theperformance of our provers
shown in Table 4. For explanations on the modeling of this example we refer to [5] pp. 55-57.

4 Outlook

This work presented in this paper has its roots in the LEO-II project (in 2006/2007 at Cam-
bridge University) in which we first studied and employed thepresented embedding of quantified
multimodal logics in simple type theory [8]6. Subsequently, we have applied the idea also to
propositional intuitionistic logics and to access controllogics [6, 9]. And similar results are
possible for various other non-classical logics that have been discussed in the literature.

For example, to model temporal logic in our framework, we mayfix particular accessibility
relations relationspast and future. Moreover, we require that both relations are transitive and
mutually inverse. This can be easily done by stating:

(transitive past) ∧ (transitive future) (42)

∀S, T (pastS T ) ⇔ (futureT S) (43)

(44)

We may then introduce further temporal operators, for example, alwaysandsometime(both
in the past, at present, and in the future):

always= λA 2pastA ∧ A ∧ 2futureA (45)

sometime= λA 3pastA ∨ A ∨ 3futureA (46)

The exploration of (quantified) temporal logic is future work. Future work also includes fur-
ther extensions of our embedding to also cover quantified hybrid logics [13, 14] and full higher-
order modal logics [17, 20]. A first suggestion in direction of higher-order modal logics has
already been made [8]. This proposal does however not yet address intensionality aspects. How-
ever, combining this proposal with non-extensional notions of models for simple type theory
[7, 21] appears a promising direction.

Our overall goal is to show that various interesting non-classical logics can be fruitfully mech-
anized and partly automated in modern proof assistants withthe help of our embedding. We also
want to motivate the integration of our automatic higher-order reasoners in modern proof assis-
tants as well as the development of further automated reasoning tools for classical higher-order
logic. It is obvious that the existing reasoners should be significantly improved for fruitful appli-
cation to challenge problems in practice. This seems not unreasonable though. Moreover, when
working with our reasoners from within a proof assistant then the user may provide interactive
help, for example, by formulating some lemmas or by splitting proof tasks in simpler subtasks.
In this context it seems also useful that our reasoners produce proof objects in a standard proof
representation format, e.g., in the TSTP format, and that a translation from this format into the
proof representations used in the proof assistants is provided.

Finally, it may be possible to formally verify the entire theory of our embedding within a
proof assistant.

6This paper was written in 2007.

A Modeling quantified multimodal logic in simple type theory

%------------------------------------------------------------------
% File : QML.ax
% Domain : Quantified multimodal logic
% Problems :
% Version :
% English : Embedding of quantified multimodal logic in
% simple type theory
% Refs :
% Source : Formalization in THF by C. Benzmueller
% Names :
% Status :
% Rating :
% Syntax :
% Comments :
%------------------------------------------------------------------
%---- declaration of additional base type mu
thf(mu,type,(

mu: $tType )).

%---- modal operators not, or, box, Pi
thf(mnot,definition,

( mnot
= ( ^ [Phi: mu > $o,W: mu] :

~ ( Phi @ W ) ) )).

thf(mor,definition,
( mor
= ( ^ [Phi: mu > $o,Psi: mu > $o,W: mu] :

( ( Phi @ W )
| ( Psi @ W ) ) ) )).

thf(mbox,definition,
( mbox
= ( ^ [R: mu > mu > $o,Phi: mu > $o,W: mu] :

! [V: mu] :
( ~ ( R @ W @ V )
| ( Phi @ V ) ) ) )).

thf(mall_ind,definition,
( mall_ind
= ( ^ [Phi: $i > mu > $o,W: mu] :

! [X: $i] :
( Phi @ X @ W ) ) )).

thf(mall_prop,definition,
( mall_prop
= ( ^ [Phi: ( mu > $o ) > mu > $o,W: mu] :

! [P: mu > $o] :
( Phi @ P @ W ) ) )).

%---- further modal operators
thf(mtrue,definition,

( mtrue
= ( ^ [W: mu] : $true ) )).

thf(mtrue,definition,
( mfalse
= ( mall_prop

@ ^ [P: mu > $o] :
( mnot @ mtrue ) ) )).

thf(mand,definition,
( mand
= ( ^ [Phi: mu > $o,Psi: mu > $o] :

( mnot @ ( mor @ ( mnot @ Phi ) @ ( mnot @ Psi ) ) ) ) )).

thf(mimpl,definition,
( mimpl
= ( ^ [Phi: mu > $o,Psi: mu > $o] :

( mor @ ( mnot @ Phi ) @ Psi ) ) )).

thf(mequiv,definition,
( mequiv
= ( ^ [Phi: mu > $o,Psi: mu > $o] :

( mand @ ( mimpl @ Phi @ Psi )
@ ( mimpl @ Psi @ Phi ) ) ) )).

thf(mdia,definition,
( mdia
= ( ^ [R: mu > mu > $o,Phi: mu > $o] :

( mnot @ ( mbox @ R @ ( mnot @ Phi ) ) ) ) )).

thf(mexi_ind,definition,
( mexi_ind
= ( ^ [Phi: $i > mu > $o] :

( mnot
@ ( mall_ind

@ ^ [X: $i] :
( mnot @ ( Phi @ X ) ) ) ) ) )).

thf(mexi_prop,definition,
( mexi_prop
= ( ^ [Phi: ( mu > $o ) > mu > $o] :

( mnot
@ ( mall_prop

@ ^ [P: mu > $o] :
( mnot @ ( Phi @ P ) ) ) ) ) )).

%------------------------------------------------------------------
%---- temporal logic

thf(past,type,(
past: mu > mu > $o )).

thf(future,type,(
future: mu > mu > $o )).

thf(past_transitive,axiom,
( mtransitive @ past )).



19

thf(future_transitive,axiom,
( mtransitive @ future )).

thf(past_future,axiom,(
! [S: mu,T: mu] :

( ( future @ S @ T )
<=> ( past @ T @ S ) ) )).

%---- definition of always
thf(malways,definition,

( malways
= ( ^ [A: mu > $o] :

( mand @ ( mand @ ( mbox @ past @ A ) @ A )
@ ( mbox @ future @ A ) ) ) )).

%---- definition of sometime
thf(msometime,definition,

( msometime
= ( ^ [A: mu > $o] :

( mor @ ( mor @ ( mdia @ past @ A ) @ A )
@ ( mdia @ future @ A ) ) ) )).

%------------------------------------------------------------------
%---- definition of validity
thf(mvalid,definition,

( mvalid
= ( ^ [Phi: mu > $o] :

! [W: mu] :
( Phi @ W ) ) )).

%---- definition of satisfiability
thf(msatisfiable,definition,

( msatisfiable
= ( ^ [Phi: mu > $o] :

? [W: mu] :
( Phi @ W ) ) )).

%---- definition of countersatisfiability
thf(mcountersatisfiable,definition,

( mcountersatisfiable
= ( ^ [Phi: mu > $o] :

? [W: mu] :
~ ( Phi @ W ) ) )).

%---- definition of invalidity
thf(minvalid,definition,

( minvalid
= ( ^ [Phi: mu > $o] :

! [W: mu] :
~ ( Phi @ W ) ) )).

%------------------------------------------------------------------
%---- definition of properties of accessibility relations
thf(mreflexive,definition,

( mreflexive

= ( ^ [R: mu > mu > $o] :
! [S: mu] :

( R @ S @ S ) ) )).

thf(msymmetric,definition,
( msymmetric
= ( ^ [R: mu > mu > $o] :

! [S: mu,T: mu] :
( ( R @ S @ T )

=> ( R @ T @ S ) ) ) )).

thf(mserial,definition,
( mserial
= ( ^ [R: mu > mu > $o] :

! [S: mu] :
? [T: mu] :

( R @ S @ T ) ) )).

thf(mtransitive,definition,
( mtransitive
= ( ^ [R: mu > mu > $o] :

! [S: mu,T: mu,U: mu] :
( ( ( R @ S @ T )

& ( R @ T @ U ) )
=> ( R @ S @ U ) ) ) )).

thf(meuclidean,definition,
( meuclidean
= ( ^ [R: mu > mu > $o] :

! [S: mu,T: mu,U: mu] :
( ( ( R @ S @ T )

& ( R @ S @ U ) )
=> ( R @ T @ U ) ) ) )).

thf(mpartially_functional,definition,
( mpartially_functional
= ( ^ [R: mu > mu > $o] :

! [S: mu,T: mu,U: mu] :
( ( ( R @ S @ T )

& ( R @ S @ U ) )
=> ( T = U ) ) ) )).

thf(mfunctional,definition,
( mfunctional
= ( ^ [R: mu > mu > $o] :

! [S: mu] :
? [T: mu] :

( ( R @ S @ T )
& ! [U: mu] :

( ( R @ S @ U )
=> ( T = U ) ) ) ) )).

thf(mweakly_dense,definition,
( mweakly_dense
= ( ^ [R: mu > mu > $o] :

! [S: mu,T: mu,U: mu] :

( ( R @ S @ T )
=> ? [U: mu] :

( ( R @ S @ U )
& ( R @ U @ T ) ) ) ) )).

thf(mweakly_connected,definition,
( mweakly_connected
= ( ^ [R: mu > mu > $o] :

! [S: mu,T: mu,U: mu] :
( ( ( R @ S @ T )
& ( R @ S @ U ) )

=> ( ( R @ T @ U )
| ( T = U )
| ( R @ U @ T ) ) ) ) )).

thf(mweakly_directed,definition,
( mweakly_directed
= ( ^ [R: mu > mu > $o] :

! [S: mu,T: mu,U: mu] :
( ( ( R @ S @ T )
& ( R @ S @ U ) )

=> ? [V: mu] :
( ( R @ T @ V )
& ( R @ U @ V ) ) ) ) )).

B A simple problem encoding

%--------------------------------------------------------------------
% File :
% Domain :
% Problems :
% Version :
% English :
% Refs :
% Source : Formalization in THF by C. Benzmueller
% Names :
% Status :
% Rating :
% Syntax :
% Comments :
%--------------------------------------------------------------------
%---- include the definitions for quantified multimodal logic
include(’QML.ax’).

%---- conjecture statement
thf(conj,conjecture,(

! [R: mu > mu > $o] :
( mvalid
@ ( mall_prop
@ ^ [A: mu > $o] :

( mall_prop
@ ^ [B: mu > $o] :

( mimpl @ ( mdia @ R @ ( mimpl @ A @ B ) )
@ ( mimpl @ ( mbox @ R @ A )

@ ( mdia @ R @ B ) ) ) ) ) ) )).

C Correspondence between accessibility relation properties and
modal axioms

%--------------------------------------------------------------------
% File :
% Domain :
% Problems :
% Version :
% English :
% Refs :
% Source : Formalization in THF by C. Benzmueller
% Names :
% Status :
% Rating :
% Syntax :
% Comments :
%--------------------------------------------------------------------
%---- include the definitions for quantified multimodal logic
include(’QML.ax’).

---- conjecture statement
thf(conj,conjecture,(

! [R: mu > mu > $o] :
( ( mvalid
@ ( mall_prop

@ ^ [A: mu > $o] :
( mimpl @ ( mbox @ R @ A )

@ ( mbox @ R @ ( mbox @ R @ A ) ) ) ) )
=> ( mtransitive @ R ) ) )).

D A simple example in epistemic logic

%--------------------------------------------------------------------
% File :
% Domain :
% Problems :
% Version :
% English :
% Refs :
% Source : Formalization in THF by C. Benzmueller
% Names :
% Status :
% Rating :
% Syntax :
% Comments :
%--------------------------------------------------------------------
%---- include the definitions for quantified multimodal logic
include(’QML.ax’).

%----
thf(peter,type,(

peter: mu > mu > $o )).

thf(john,type,(
john: mu > mu > $o )).

thf(wife,type,(
wife: ( mu > mu > $o ) > mu > mu > $o )).

%----
thf(refl_peter,axiom,

( mreflexive @ peter )).

thf(refl_john,axiom,
( mreflexive @ john )).

thf(refl_wife_peter,axiom,
( mreflexive @ ( wife @ peter ) )).

thf(trans_peter,axiom,
( mtransitive @ peter )).

thf(trans_john,axiom,
( mtransitive @ john )).

thf(trans_wife_peter,axiom,
( mtransitive @ ( wife @ peter ) )).

%----
thf(ax_i,axiom,

( mvalid
@ ( mall_prop

@ ^ [A: mu > $o] :
( mimpl @ ( mbox @ peter @ ( mbox @ john @ A ) )

@ ( mbox @ john @ ( mbox @ peter @ A ) ) ) ) )).

thf(ax_ii,axiom,
( mvalid
@ ( mall_prop

@ ^ [A: mu > $o] :
( mimpl @ ( mbox @ ( wife @ peter ) @ A )

@ ( mbox @ peter @ A ) ) ) )).

thf(time,type,(
time: mu > $o )).

thf(place,type,(
place: mu > $o )).

thf(appointment,type,(
appointment: mu > $o )).

thf(ax_a,axiom,
( mvalid @ ( mbox @ peter @ time ) )).

thf(ax_b,axiom,
( mvalid @ ( mbox @ peter @ ( mbox @ john @ place ) ) )).

thf(ax_c,axiom,
( mvalid @ ( mbox @ ( wife @ peter )

@ ( mimpl @ ( mbox @ peter @ time )
@ ( mbox @ john @ time ) ) ) )).

thf(ax_d,axiom,
( mvalid @

( mbox @ peter
@ ( mbox @ john

@ ( mimpl @ ( mand @ place @ time )
@ appointment ) ) ) )).

thf(conj,conjecture,
( mvalid @

( mand @ ( mbox @ peter
@ ( mbox @ john @ appointment ) )

@ ( mbox @ john
@ ( mbox @ peter @ appointment ) ) ) )).

E Wise men puzzle

%--------------------------------------------------------------------
% File :
% Domain :
% Problems : Wise men puzzle
% Version :
% English :
% Refs : Matteo Baldoni, Normal Multimodal Logics: Automatic
% Deduction and Logic Programming Extension, Phd thesis,
% Universita degli studi di Torino, 2003, pp.55-57
% Source : Formalization in THF by C. Benzmueller
% Names :
% Status :
% Rating :
% Syntax :
% Comments :
%--------------------------------------------------------------------
%---- include the definitions for quantified multimodal logic
include(’QML.ax’).

%----
thf(a,type,(

a: mu > mu > $o )).

thf(b,type,(
b: mu > mu > $o )).

thf(c,type,(
c: mu > mu > $o )).

thf(fool,type,(
fool: mu > mu > $o )).

thf(ws,type,(
ws: ( mu > mu > $o ) > mu > $o )).

%----
thf(axiom_1,axiom,
( mvalid @ ( mbox @ fool @

( mor @ ( ws @ a ) @ ( mor @ ( ws @ b ) @ ( ws @ c ) ) ) ) )).

%----
thf(axiom_2_a_b,axiom,
( mvalid @ ( mbox @ fool @

( mimpl @ ( ws @ a ) @ ( mbox @ b @ ( ws @ a ) ) ) ) )).

thf(axiom_2_a_c,axiom,
( mvalid @ ( mbox @ fool @

( mimpl @ ( ws @ a ) @ ( mbox @ c @ ( ws @ a ) ) ) ) )).

thf(axiom_2_b_a,axiom,
( mvalid @ ( mbox @ fool @

( mimpl @ ( ws @ b ) @ ( mbox @ a @ ( ws @ b ) ) ) ) )).

thf(axiom_2_b_c,axiom,

( mvalid @ ( mbox @ fool @
( mimpl @ ( ws @ b ) @ ( mbox @ c @ ( ws @ b ) ) ) ) )).

thf(axiom_2_c_a,axiom,
( mvalid @ ( mbox @ fool @
( mimpl @ ( ws @ c ) @ ( mbox @ a @ ( ws @ c ) ) ) ) )).

thf(axiom_2_b_a,axiom,
( mvalid @ ( mbox @ fool @
( mimpl @ ( ws @ c ) @ ( mbox @ b @ ( ws @ c ) ) ) ) )).

%----
thf(axiom_3_a_b,axiom,
( mvalid @ ( mbox @ fool @
( mimpl @ ( mnot @ ( ws @ a ) )

@ ( mbox @ b @ ( mnot @ ( ws @ a ) ) ) ) ) )).

thf(axiom_3_a_c,axiom,
( mvalid @ ( mbox @ fool @
( mimpl @ ( mnot @ ( ws @ a ) )

@ ( mbox @ c @ ( mnot @ ( ws @ a ) ) ) ) ) )).

thf(axiom_3_b_a,axiom,
( mvalid @ ( mbox @ fool @
( mimpl @ ( mnot @ ( ws @ b ) )

@ ( mbox @ a @ ( mnot @ ( ws @ b ) ) ) ) ) )).

thf(axiom_3_b_c,axiom,
( mvalid @ ( mbox @ fool @
( mimpl @ ( mnot @ ( ws @ b ) )

@ ( mbox @ c @ ( mnot @ ( ws @ b ) ) ) ) ) )).

thf(axiom_3_c_a,axiom,
( mvalid @ ( mbox @ fool @
( mimpl @ ( mnot @ ( ws @ c ) )

@ ( mbox @ a @ ( mnot @ ( ws @ c ) ) ) ) ) )).

thf(axiom_3_b_a,axiom,
( mvalid @ ( mbox @ fool @
( mimpl @ ( mnot @ ( ws @ c ) )

@ ( mbox @ b @ ( mnot @ ( ws @ c ) ) ) ) ) )).

%----
thf(t_axiom_for_fool,axiom,

( mvalid
@ ( mall_prop

@ ^ [A: mu > $o] :
( mimpl @ ( mbox @ fool @ A ) @ A ) ) )).

thf(k_axiom_for_fool,axiom,
( mvalid
@ ( mall_prop

@ ^ [A: mu > $o] :
( mimpl @ ( mbox @ fool @ A )

@ ( mbox @ fool @ ( mbox @ fool @ A ) ) ) ) )).

thf(i_axiom_for_fool_a,axiom,
( mvalid
@ ( mall_prop

@ ^ [Phi: mu > $o] :
( mimpl @ ( mbox @ fool @ Phi )

@ ( mbox @ a @ Phi ) ) ) )).

thf(i_axiom_for_fool_b,axiom,
( mvalid
@ ( mall_prop

@ ^ [Phi: mu > $o] :
( mimpl @ ( mbox @ fool @ Phi )

@ ( mbox @ b @ Phi ) ) ) )).

thf(i_axiom_for_fool_c,axiom,
( mvalid
@ ( mall_prop

@ ^ [Phi: mu > $o] :
( mimpl @ ( mbox @ fool @ Phi )

@ ( mbox @ c @ Phi ) ) ) )).

%----
thf(a7_axiom_for_fool_a_b,axiom,

( mvalid
@ ( mall_prop

@ ^ [Phi: mu > $o] :
( mimpl @ ( mbox @ a @ Phi )

@ ( mbox @ b @ ( mbox @ a @ Phi ) ) ) ) )).

thf(a7_axiom_for_fool_a_c,axiom,
( mvalid
@ ( mall_prop

@ ^ [Phi: mu > $o] :
( mimpl @ ( mbox @ a @ Phi )

@ ( mbox @ c @ ( mbox @ a @ Phi ) ) ) ) )).

thf(a7_axiom_for_fool_b_a,axiom,
( mvalid
@ ( mall_prop

@ ^ [Phi: mu > $o] :
( mimpl @ ( mbox @ b @ Phi )

@ ( mbox @ a @ ( mbox @ b @ Phi ) ) ) ) )).

thf(a7_axiom_for_fool_b_c,axiom,
( mvalid
@ ( mall_prop

@ ^ [Phi: mu > $o] :
( mimpl @ ( mbox @ b @ Phi )

@ ( mbox @ c @ ( mbox @ b @ Phi ) ) ) ) )).

thf(a7_axiom_for_fool_c_a,axiom,
( mvalid
@ ( mall_prop

@ ^ [Phi: mu > $o] :
( mimpl @ ( mbox @ c @ Phi )

@ ( mbox @ a @ ( mbox @ c @ Phi ) ) ) ) )).



29

thf(a7_axiom_for_fool_c_b,axiom,
( mvalid
@ ( mall_prop

@ ^ [Phi: mu > $o] :
( mimpl @ ( mbox @ c @ Phi )

@ ( mbox @ b @ ( mbox @ c @ Phi ) ) ) ) )).

%----
thf(a6_axiom_for_fool_a_b,axiom,

( mvalid
@ ( mall_prop

@ ^ [Phi: mu > $o] :
( mimpl @ ( mnot @ ( mbox @ a @ Phi ) )

@ ( mbox @ b @ ( mnot @ ( mbox @ a @ Phi ) ) ) ) ) )).

thf(a6_axiom_for_fool_a_c,axiom,
( mvalid
@ ( mall_prop

@ ^ [Phi: mu > $o] :
( mimpl @ ( mnot @ ( mbox @ a @ Phi ) )

@ ( mbox @ c @ ( mnot @ ( mbox @ a @ Phi ) ) ) ) ) )).

thf(a6_axiom_for_fool_b_a,axiom,
( mvalid
@ ( mall_prop

@ ^ [Phi: mu > $o] :
( mimpl @ ( mnot @ ( mbox @ b @ Phi ) )

@ ( mbox @ a @ ( mnot @ ( mbox @ b @ Phi ) ) ) ) ) )).

thf(a6_axiom_for_fool_b_c,axiom,
( mvalid
@ ( mall_prop

@ ^ [Phi: mu > $o] :
( mimpl @ ( mnot @ ( mbox @ b @ Phi ) )

@ ( mbox @ c @ ( mnot @ ( mbox @ b @ Phi ) ) ) ) ) )).

thf(a6_axiom_for_fool_c_a,axiom,
( mvalid
@ ( mall_prop

@ ^ [Phi: mu > $o] :
( mimpl @ ( mnot @ ( mbox @ c @ Phi ) )

@ ( mbox @ a @ ( mnot @ ( mbox @ c @ Phi ) ) ) ) ) )).

thf(a6_axiom_for_fool_c_b,axiom,
( mvalid
@ ( mall_prop

@ ^ [Phi: mu > $o] :
( mimpl @ ( mnot @ ( mbox @ c @ Phi ) )

@ ( mbox @ b @ ( mnot @ ( mbox @ c @ Phi ) ) ) ) ) )).

%----
thf(axiom_4,axiom,

( mvalid @ ( mnot @ ( mbox @ a @ ( ws @ a ) ) ) )).

thf(axiom_5,axiom,

( mvalid @ ( mnot @ ( mbox @ b @ ( ws @ b ) ) ) )).

%----
thf(conj,conjecture,

( mvalid @ ( mbox @ c @ ( ws @ c ) ) )).


